/**
 * Copyright (c) 2010, iDA MediaFoundry
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *
 * - Redistributions of source code must retain the above copyright notice,
 *   this list of conditions and the following disclaimer.
 *
 * - Redistributions in binary form must reproduce the above copyright notice,
 *   this list of conditions and the following disclaimer in the documentation
 *   and/or other materials provided with the distribution.
 *
 * - Neither the name of iDA MediaFoundry nor the names of its contributors may
 *   be used to endorse or promote products derived from this software without
 *   specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 */
package be.idamf.cortizone.ioc.container
{
	import be.idamf.flash.events.dispatchEventOn;
	
	import flash.events.Event;
	
	import org.floxy.ProxyRepository;
	
	[Event(name="complete", type="flash.events.Event")]
	
	/**
	 * A BeanProxy will create an object that is of the specified Class or 
	 * Interface type.
	 * 
	 * Method invocation on this object will be intercepted and delegated to
	 * the target. This is typically a dynamic flash.utils.Proxy object.
	 * 
	 * Benefit: you code against an interface but you don't need to write the
	 * implementation which is runtime/dynamic.
	 * 
	 * @author Jan Van Coppenolle
	 */
	public class BeanProxy extends AbstractAsyncBean
	{
		/**
		 * @private
		 */
		public static var proxyRepo:ProxyRepository = new ProxyRepository();
		
		////////////////////////////////////////////////////////////////////////
		// Instance
		
		[Bindable]
		/**
		 * The proxy target.
		 */
		public var target:Object;
		
		////////////////////////////////////////////////////////////////////////
		// Constructor
		
		/**
		 * @private
		 */
		public function BeanProxy()
		{
			super();
		}
		
		////////////////////////////////////////////////////////////////////////
		// Accessors
		
		private var _type:Class;
		[Bindable]
		/**
		 * The type of this bean.
		 */
		public function get type():Class
		{
			return _type;
		}
		public function set type(value:Class):void
		{
			_type = value;
			proxyRepo.prepare([_type]).addEventListener(Event.COMPLETE, proxyRepo_completeHandler);
		}
		
		////////////////////////////////////////////////////////////////////////
		// Handlers
		
		/**
		 * @private
		 */
		protected function proxyRepo_completeHandler(event:Event):void
		{
			_value = proxyRepo.create(_type, [], new ProxyInterceptor(target));
			_isComplete = true;
			dispatchEventOn(this, Event.COMPLETE);
		}
	}
}

import org.floxy.IInterceptor;
import org.floxy.IInvocation;

internal class ProxyInterceptor implements IInterceptor
{
	private var target:Object;
	
	public function ProxyInterceptor(target:Object)
	{
		this.target = target;
	}
	
	public function intercept(invocation:IInvocation):void
	{
		invocation.returnValue = (target[invocation.method.name] as Function).apply(this, invocation.arguments);
	}
}